Séries de Fourier

Podemos usar o Maple para calcular os coeficientes de Fourier e ilustar a convergência das séries de Fourier.

Calculamos os coeficientes da série de Fourier de uma função de x definida sobre um intervalo de **-p** a **p** usando o seguinte procedimento, onde f é a função, n o número de termos da série de fourier e p é o periodo.

```
> restart:
```

```
> acoeff:=(f,n,p)->(1/p)*evalf(Int(f(t)*cos(n*Pi*t/p),t=-p..p));
```

$$acoeff := (f, n, p) \rightarrow \frac{\mathrm{evalf}\left(\int_{-p}^{p} f(t) \cos\left(\frac{n \pi t}{p}\right) dt\right)}{p}$$

```
> bcoeff:=(f,n,p)->(1/p)*evalf(Int(f(t)*sin(n*Pi*t/p),t=-p..p));
```

$$bcoeff := (f, n, p) \rightarrow \frac{\mathrm{evalf}\left(\int_{-p}^{p} f(t) \sin\left(\frac{n \pi t}{p}\right) dt\right)}{p}$$

```
> fourierseries:=proc(f,n,p) local s,ii;

> s:=acoeff(f,0,p)/2:

> for ii from 1 to n do s:=s+acoeff(f,ii,p)*cos(ii*Pi*x/p)+bcoeff(f,ii,p)*sin(ii*Pi*x/p) od:

> s;

> end:
```

> **f:=x->x^2;**

$$f := x \rightarrow x^2$$

> **s:=fourierseries (f,20,Pi);**

$$s := \frac{10.33542556}{\pi} - 12.56637061\,\frac{\cos(x)}{\pi} + 3.141592654\,\frac{\cos(2\,x)}{\pi} - 1.396263402\,\frac{\cos(3\,x)}{\pi}$$
$$+ .7853981634\,\frac{\cos(4\,x)}{\pi} - .5026548246\,\frac{\cos(5\,x)}{\pi} + .3490658504\,\frac{\cos(6\,x)}{\pi} - .2564565432\,\frac{\cos(7\,x)}{\pi}$$
$$+ .1963495409\,\frac{\cos(8\,x)}{\pi} - .1551403780\,\frac{\cos(9\,x)}{\pi} + .1256637061\,\frac{\cos(10\,x)}{\pi} - .1038543026\,\frac{\cos(11\,x}{\pi}$$
$$+ .08726646260\,\frac{\cos(12\,x)}{\pi} - .07435722257\,\frac{\cos(13\,x)}{\pi} + .06411413579\,\frac{\cos(14\,x)}{\pi} - .05585053607\,\frac{c}{}$$
$$+ .04908738522\,\frac{\cos(16\,x)}{\pi} - .04348225126\,\frac{\cos(17\,x)}{\pi} + .03878509449\,\frac{\cos(18\,x)}{\pi} - .03480989090\,\frac{c}{}$$
$$+ .03141592654\,\frac{\cos(20\,x)}{\pi}$$

> **plot({s,f(x)},x=-Pi..Pi);**



Vamos olhar em vários períodos! Temos um truque para plotar a extensão periódica de x^2. Primeiro precisamos uma expressão periódica da função identidade x:

> **pex:=(x,p)->-p+ x+p-floor((x+p)/(2*p))*2*p;**

$$pex := (x,p) \rightarrow x - 2\,\mathrm{floor}\left(\frac{1}{2}\,\frac{x+p}{p}\right)p$$

> **plot(pex(x,5),x=-20..20);**



Então podemos substituir pex para x em **x^2** :

> **plot({s,subs(x=pex(x,Pi),x^2)},x=-10..10);**



O que acontece na descontinuidade das funcoes, por exemplo pex ?

(Note que podemos calcular a funcao **x->x** quando calculamos a série de Fourier, pois esta coincide com **pex(x,Pi)** sobre o intervalo de intergraçao, e porque fazendo isto economizamos tempo de computaçao.

```
> plot({pex(x,Pi),fourierseries(x->x,1,Pi)},x=-10..10);
```



```
> plot({pex(x,Pi),fourierseries(x->x,10,Pi)},x=-10..10);
```




```
>
```